```
> with(OreModules):
> with(OreMorphisms):
> with(Stafford):
> with(linalg):
```

Let us consider the second Weyl algebra $A = A_2(\mathbb{Q})$, where $\mathbb{Q}$ is the field of rational numbers,

```
> A := DefineOreAlgebra(diff=[dx,x], diff=[dy,y], polynom=[x,y]):
```

and the left A-module M finitely presented by the following matrix:

```
> R := evalm([[dx,dy,0,0,0,0],[0,1,-1,0,dx,dy],[0,0,dx,dy,0,0]]);
```

$$R := \begin{bmatrix} dx & dy & 0 & 0 & 0 & 0 \\ 0 & 1 & -1 & 0 & dx & dy \\ 0 & 0 & dx & dy & 0 & 0 \end{bmatrix}$$

The left A-module M corresponds to *Cosserat's equations* appearing in linear elasticity.

The rank of M is:

```
> OreRank(R,A);
```

$$3$$

Thus, M admits a unimodular element. Let us compute one:

```
> U := UnimodularElement(R,A);
```

$$U := \left[\begin{bmatrix} 0 & 0 & 0 & 0 & 1 & 0 \end{bmatrix}, \begin{bmatrix} dy \\ -dx \\ 0 \\ 0 \\ 1 \\ 0 \end{bmatrix} \right]$$

The residue class of U[1] is a unimodular element $m_1{*}$ of M. Moreover, the left A-homomorphism ϕ_1 from M to A induced by U[2] satisfies $\phi_1(m_1{*}) =$ U[1] U[2] $= 1$:

```
> Mult(U[1],U[2],A);
```

$$\begin{bmatrix} 1 \end{bmatrix}$$

Let us check that ϕ_1 is a well-defined left A-homomorphism from M to A, i.e., U[2] in ker_A(R.):

```
> Mult(R,U[2],A);
```

$$\begin{bmatrix} 0 \\ 0 \\ 0 \end{bmatrix}$$

Therefore, M is the direct sum of $A\, m_1{*}$ and $\ker(\phi_1)$. The left A-submodule $\ker(\phi_1)$ of M can be computed using the command *FreeDirectSummand* with the option "kernel".

```
> E := FreeDirectSummand(R,A,"kernel");
```

$$E := \left[\begin{bmatrix} 0 & 0 & 0 & 0 & 1 & 0 \end{bmatrix}, \begin{bmatrix} dy \\ -dx \\ 0 \\ 0 \\ 1 \\ 0 \end{bmatrix}, \begin{bmatrix} -1 & 0 & 0 & 0 & dy & 0 \\ 0 & 0 & 1 & 0 & 0 & -dy \\ 0 & 0 & 1 & 0 & 0 & 0 \\ 0 & 0 & 0 & 1 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 1 \end{bmatrix} \right]$$

The first entry E[1] of E corresponds to U[1] and the second entry E[2] is U[2]. The residue classes of the rows of the third entry E[3] of E generate $\ker(\phi_1)$.

The command *FreeDirectSummand* with the option "presentation" computes a presentation of $\ker(\phi_1)$.

```
> F := FreeDirectSummand(R,A,"presentation");
```

$$F := \left[\begin{bmatrix} 0 & 0 & 0 & 0 & 1 & 0 \end{bmatrix}, \begin{bmatrix} dy \\ -dx \\ 0 \\ 0 \\ 1 \\ 0 \end{bmatrix}, \begin{bmatrix} 0 & 1 & 0 & 0 & 0 & 0 \\ 0 & 0 & 1 & -1 & 0 & dy \\ 0 & 0 & 0 & dx & dy & 0 \\ dx & 1 & -dy & 0 & 0 & 0 \end{bmatrix}, \begin{bmatrix} -1 & 0 & 0 & 0 & dy & 0 \\ dx & dy & 0 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 & dx & 0 \\ 0 & 0 & 1 & 0 & 0 & 0 \\ 0 & 0 & 0 & 1 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 1 \end{bmatrix} \right]$$

The first entry F[1] of F corresponds to U[1] and the second entry F[2] of F is U[2]. The third entry F[3] of F is a presentation of $\ker(\phi_1)$. Finally, F[4] induces a left A-homomorphism i_1 from this finitely presented left A-module O_1 to M. Let us check again that i_1 is injective:

```
> TestInj(F[3],R,F[4],A);
```

$$true$$

Using the option "isomorphism" of the command *FreeDirectSummand*

```
> F2 := FreeDirectSummand(R,A,"isomorphism");
```

$$F2 := \left[\begin{bmatrix} 0 & 0 & 0 & 0 & 1 & 0 \end{bmatrix}, \begin{bmatrix} dy \\ -dx \\ 0 \\ 0 \\ 1 \\ 0 \end{bmatrix}, \begin{bmatrix} 0 & 0 & 1 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 1 & -1 & 0 & dy \\ 0 & 0 & 0 & 0 & dx & dy & 0 \\ 0 & dx & 1 & -dy & 0 & 0 & 0 \end{bmatrix}, \begin{bmatrix} 0 & 0 & 0 & 0 & 1 & 0 \\ -1 & 0 & 0 & 0 & dy & 0 \\ dx & dy & 0 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 & dx & 0 \\ 0 & 0 & 1 & 0 & 0 & 0 \\ 0 & 0 & 0 & 1 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 1 \end{bmatrix} \right]$$

we first obtain a representative F2[1] of the unimodular element $m_1{}^*$ of M and a left A-homomorphism ϕ_1 from M to A induced by F2[2] such that $\phi_1(m_1{}^*) = 1$. Moreover, the left A-module M_1 finitely presented by the third entry F2[3] of F2 is such that M_1 is isomorphic to the direct sum of A and $\ker(\phi_1)$. The left A-homomorphism g_1 from M_1 to M induced by F2[4], where F2[4] is the fourth entry of F2, is a left A-isomorphism:

```
> TestIso(F2[3],R,F2[4],A);
```

$$true$$

Let us compute the rank of the left A-module $\ker(\phi_1)$:

```
> OreRank(F[3],A);
```

$$2$$

Since $rank_A(\ker(\phi_1)) = 2$, there exists a unimodular element of the left A-module $\ker(\phi_1)$.

```
>  G := FreeDirectSummand(F[3],A,"presentation");
```

$$G := \left[\left[\begin{array}{cccccc} 0 & 0 & 0 & 0 & 0 & 1 \end{array}\right], \left[\begin{array}{c} 0 \\ 0 \\ 0 \\ dy \\ -dx \\ 1 \end{array}\right], \left[\begin{array}{cccccc} 0 & 1 & 0 & 0 & 0 & 0 \\ 0 & 0 & 1 & 1 & 0 & 0 \\ 0 & 0 & 0 & 0 & 1 & 0 \\ dx & 1 & -dy & 0 & 0 & 0 \\ 0 & 0 & 0 & dx & 1 & -dy \end{array}\right], \left[\begin{array}{cccccc} 1 & 0 & 0 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 & 0 & 0 \\ 0 & 0 & 1 & 0 & 0 & 0 \\ 0 & 0 & 0 & -1 & 0 & dy \\ 0 & 0 & 0 & dx & dy & 0 \\ 0 & 0 & 0 & 0 & 1 & dx \end{array}\right] \right]$$

We have that $\ker(\phi_1)$ is the direct sum of $A\ m_2$ and $\ker(\phi_2)$, where m_2* is the unimodular element of $\ker(\phi_1)$ represented by G[1] and ϕ_2 is the left A-homomorphism from $\ker(\phi_1)$ to A induced by G[2] which satisfies $\phi_2(m_2*) =$ G[1] G[2] $= 1$. Moreover, $\ker(\phi_2)$ is isomorphic to the left A-module finitely presented by G[3], and the embedding i_2 from this left A-module O_2 to $\ker(\phi_1)$ is induced by G[4]. Let us check again that i_2 is injective:

```
>  TestInj(G[3],F[3],G[4],A);
```

$$true$$

Now, the rank of O_2 is equal to:

```
>  OreRank(G[3],A);
```

$$1$$

Hence, the above technique cannot be applied again to the left A-module $\ker(\phi_2)$.

The left A-homomorphism $\iota = i_2 \text{ o } i_1$ from O_2 to M is then induced by J, where J = G[4] F[4] is defined by:

```
>  J := Mult(G[4],F[4],A);
```

$$J := \left[\begin{array}{cccccc} -1 & 0 & 0 & 0 & dy & 0 \\ dx & dy & 0 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 & dx & 0 \\ 0 & 0 & -1 & 0 & 0 & dy \\ 0 & 0 & dx & dy & 0 & 0 \\ 0 & 0 & 0 & 1 & 0 & dx \end{array}\right]$$

Let us check again that ι is injective:

```
>  TestInj(G[3],R,J,A);
```

$$true$$

Elementary operations can be used to simplify the presentation G[3] of O_2.

```
>  with(PurityFiltration):
>  K := ReducedPresentation(G[3],A);
```

$$K := \left[\left[\begin{array}{cccccc} 0 & 1 & 0 & 0 & 0 & 0 \\ 0 & 0 & 1 & 1 & 0 & 0 \\ 0 & 0 & 0 & 0 & 1 & 0 \\ dx & 1 & -dy & 0 & 0 & 0 \\ 0 & 0 & 0 & dx & 1 & -dy \end{array}\right], \left[\begin{array}{ccc} dx & 0 & -dy \\ dy & dx & 0 \end{array}\right], \left[\begin{array}{ccc} 0 & 1 & 0 \\ 0 & 0 & 0 \\ -1 & 0 & 0 \\ 1 & 0 & 0 \\ 0 & 0 & 0 \\ 0 & 0 & 1 \end{array}\right], \left[\begin{array}{cccccc} 0 & 0 & 0 & 1 & 0 & 0 \\ 1 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 1 \end{array}\right] \right]$$

We obtain that S = K[2] defined by

```
>  S := K[2];
```

$$S := \begin{bmatrix} dx & 0 & -dy \\ dy & dx & 0 \end{bmatrix}$$

is a presentation matrix of a left A-module O$_2$' which is isomorphic to O$_2$. Moreover, the left A-homomorphism j_2 from O$_2$' to O$_2$ induced by K[4] is a left A-isomorphism. Thus, we get that M is isomorphic to the direct sum of $A^{1\times 2}$ and O$_2$'. The left A-homomorphism χ from O$_2$' to M induced by P, where P = K[4] J is defined by

```
>  P := Mult(K[4],J,A);
```

$$P := \begin{bmatrix} 0 & 0 & -1 & 0 & 0 & dy \\ -1 & 0 & 0 & 0 & dy & 0 \\ 0 & 0 & 0 & 1 & 0 & dx \end{bmatrix}$$

is injective:

```
>  TestInj(S,R,P,A);
```

$$\mathit{true}$$

We note that the left A-module O$_2$' corresponds to the linear PD system defining the equilibrium of the symmetric stress tensor. Hence, if F is a left A-module, then ker_F(R.) is isomorphic to the direct sum of F^2 and ker_F(S.), which shows that the solution space of Cosserat's equations is isomorphic to the direct sum of F^2 and the solutions of the classical linear PD system defining the equilibrium of the symmetric stress tensor. Since x and y do not appear in the coefficients of the unimodular elements and in their corresponding forms, the above results are also valid over the commutative polynomial ring B = $\mathbb{Q}$[dx,dy] and for any B-module F.

Using the command *MaximalFreeDirectSummand*, the decomposition of M as a direct sum of $A^{1\times 2}$ and O$_2$ can be obtained in one step.

```
>  N := MaximalFreeDirectSummand(R,A,"presentation");
```

$$N := \left[\left[\left[\begin{bmatrix} 0 & 0 & 0 & 0 & 1 & 0 \end{bmatrix}, \begin{bmatrix} dy \\ -dx \\ 0 \\ 0 \\ 1 \\ 0 \end{bmatrix}\right], \left[\begin{bmatrix} 0 & 0 & 0 & 0 & 0 & 1 \end{bmatrix}, \begin{bmatrix} 0 \\ 0 \\ dy \\ -dx \\ 0 \\ 1 \end{bmatrix}\right]\right], \begin{bmatrix} 0 & 1 & 0 & 0 & 0 & 0 \\ 0 & 0 & 1 & 1 & 0 & 0 \\ 0 & 0 & 0 & 0 & 1 & 0 \\ dx & 1 & -dy & 0 & 0 & 0 \\ 0 & 0 & 0 & dx & 1 & -dy \end{bmatrix}, \begin{bmatrix} -1 & 0 & 0 & 0 & dy & 0 \\ dx & dy & 0 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 & dx & 0 \\ 0 & 0 & -1 & 0 & 0 & dy \\ 0 & 0 & dx & dy & 0 & 0 \\ 0 & 0 & 0 & 1 & 0 & dx \end{bmatrix}\right]$$

The first entry N[1] of N returns two unimodular elements m_1* and m_2* of M and the corresponding two forms.

```
>  N[1];
```

$$\left[\left[\left[\begin{array}{cccccc} 0 & 0 & 0 & 0 & 1 & 0 \end{array}\right], \left[\begin{array}{c} dy \\ -dx \\ 0 \\ 0 \\ 1 \\ 0 \end{array}\right]\right], \left[\left[\begin{array}{cccccc} 0 & 0 & 0 & 0 & 0 & 1 \end{array}\right], \left[\begin{array}{c} 0 \\ 0 \\ dy \\ -dx \\ 0 \\ 1 \end{array}\right]\right]\right]$$

The second entry N[2], namely,

```
> N[2];
```

$$\left[\begin{array}{cccccc} 0 & 1 & 0 & 0 & 0 & 0 \\ 0 & 0 & 1 & 1 & 0 & 0 \\ 0 & 0 & 0 & 0 & 1 & 0 \\ dx & 1 & -dy & 0 & 0 & 0 \\ 0 & 0 & 0 & dx & 1 & -dy \end{array}\right]$$

is a presentation matrix of the left A-module O_2 which is isomorphic to $\ker(\phi_2)$. Moreover, the left A-homomorphism i_2 from O_2 to M induced by N[3], where the matrix N[3] is defined by

```
> N[3];
```

$$\left[\begin{array}{cccccc} -1 & 0 & 0 & 0 & dy & 0 \\ dx & dy & 0 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 & dx & 0 \\ 0 & 0 & -1 & 0 & 0 & dy \\ 0 & 0 & dx & dy & 0 & 0 \\ 0 & 0 & 0 & 1 & 0 & dx \end{array}\right]$$

is injective:

```
> TestInj(N[2],R,N[3],A);
```

$$true$$

Since the rank of O_2 is one

```
> OreRank(N[2],A);
```

$$1$$

the above technique cannot be applied again to decompose O_2.

If we use the option "isomorphism" of the command *MaximalFreeDirectSummand*,

```
> N2 := MaximalFreeDirectSummand(R,A,"isomorphism"):
```

then we first find again two unimodular elements $m_1{}^*$ and $m_2{}^*$ of M with their corresponding forms ϕ_1 and ϕ_2,

```
> N2[1];
```

$$\left[\left[\left[\begin{array}{cccccc} 0 & 0 & 0 & 0 & 1 & 0 \end{array}\right], \left[\begin{array}{c} dy \\ -dx \\ 0 \\ 0 \\ 1 \\ 0 \end{array}\right]\right], \left[\left[\begin{array}{cccccc} 0 & 0 & 0 & 0 & 0 & 1 \end{array}\right], \left[\begin{array}{c} 0 \\ 0 \\ dy \\ -dx \\ 0 \\ 1 \end{array}\right]\right]\right]$$

and the left A-module M_1 finitely presented by the second entry N2[2] of N2, namely,

```
>  N2[2];
```

$$\begin{bmatrix} 0 & 0 & 0 & 1 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 1 & 1 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 1 & 0 \\ 0 & 0 & dx & 1 & -dy & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & dx & 1 & -dy \end{bmatrix}$$

is isomorphic to the direct sum of $A^{1\times 2}$ and O_2. The left A-isomorphism g from M_1 to M is induced by N2[3], where the third entry N2[3] of N2 is defined by:

```
>  N2[3];
```

$$\begin{bmatrix} 0 & 0 & 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 0 & 0 & 1 \\ -1 & 0 & 0 & 0 & dy & 0 \\ dx & dy & 0 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 & dx & 0 \\ 0 & 0 & -1 & 0 & 0 & dy \\ 0 & 0 & dx & dy & 0 & 0 \\ 0 & 0 & 0 & 1 & 0 & dx \end{bmatrix}$$

We can check again that g is a left A-isomorphism:

```
>  TestIso(N2[2],R,N2[3],A);
```

$$true$$

Let us compute $f = g^{-1}$. We have that f is induced by X[1], where X[1] is the first entry of X defined by:

```
>  X := InverseMorphism(N2[2],R,N2[3],A);
```

$$X := \left[\begin{bmatrix} dy & 0 & -1 & 0 & 0 & 0 & 0 & 0 \\ -dx & 0 & 0 & 0 & 0 & -1 & 0 & 0 \\ 0 & dy & 0 & 0 & 0 & -1 & 0 & 0 \\ 0 & -dx & 0 & 0 & 0 & 0 & 0 & 1 \\ 1 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 & 0 & 0 & 0 & 0 \end{bmatrix}, \begin{bmatrix} 1 & -dy & 0 & -1 & 0 \\ 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 1 & 0 & -1 \end{bmatrix}\right]$$

```
>  TestIso(R,N2[2],X[1],A);
```

$$true$$